

# Amortissement d'une immobilisation

Implémentation dans Excel

Jean-Marie Lambert

Septembre 2017

## 1 Définitions

### 1.1 Amortissement linéaire

#### L'amortissement linéaire

Il consiste à répartir de manière égale les amortissements sur la durée de vie de l'immobilisation.

- Lorsqu'une immobilisation est mise en service en cours d'année,
- la première annuité doit être réduite au prorata de la durée entre la date de mise en service et la date de clôture de l'exercice,
- elle se calcule en jours et sur une durée de 360 jours pour un exercice de 12 mois,
- la dernière annuité s'en retrouve également ajustée.

### 1.2 Amortissement dégressif

#### L'amortissement dégressif

Il permet d'amortir fortement un bien en début d'utilisation.

- La première année on amortit au taux double de l'amortissement linéaire (max : 40%),
- la première annuité doit être réduite au prorata de la durée entre la date de mise en service et la date de clôture de l'exercice,
- les années suivantes, on amortit le solde au même taux double jusqu'au moment où le taux linéaire donne un montant supérieur.

## 2 Exemples

### 2.1 Amortissement linéaire

	A	B	C	D	E	F
1	<b>AMORTISSEMENT LINEAIRE</b>					
2	Données					
3		Valeur du bien	10 000,00			
4		Nb années	10			
5		Date d'acquisition	15/07/2016	11		
6		Nb jours 1 <sup>re</sup> année	166	461,11		
7		Taux	10,00%	1 000,00		
8						
9						
10		Année	Base à amortir	Amortissement	Cumul	Valeur résiduelle
11		0				10 000,00
12		1	10 000,00	461,11	461,11	9 538,89
13		2	10 000,00	1 000,00	1 461,11	8 538,89
14		3	10 000,00	1 000,00	2 461,11	7 538,89
15		4	10 000,00	1 000,00	3 461,11	6 538,89
16		5	10 000,00	1 000,00	4 461,11	5 538,89
17		6	10 000,00	1 000,00	5 461,11	4 538,89
18		7	10 000,00	1 000,00	6 461,11	3 538,89
19		8	10 000,00	1 000,00	7 461,11	2 538,89
20		9	10 000,00	1 000,00	8 461,11	1 538,89
21		10	10 000,00	1 000,00	9 461,11	538,89
22		11	10 000,00	538,89	10 000,00	0,00

## 2.2 Amortissement dégressif

A	B	C	D	E	F
	<b>AMORTISSEMENT DÉGRESSIF</b>				
	à double taux				
1					
2	Données				
3	Valeur du bien		10 000,00		
4	Nb années		10		
5	Date d'acquisition		15/07/2016		
6	Nb jours 1 <sup>re</sup> année		166		
7	Taux dégressif		20,00%		
8	Taux linéaire		10,00%		
9					
10		Base à amortir	Amortissement	Cumul	Valeur résiduelle
11	Année				
12	1	10 000,00	922,22	922,22	9 077,78
13	2	9 077,78	1 815,56	2 737,78	7 262,22
14	3	7 262,22	1 452,44	4 190,22	5 809,78
15	4	5 809,78	1 161,96	5 352,18	4 647,82
16	5	4 647,82	1 000,00	6 352,18	3 647,82
17	6	3 647,82	1 000,00	7 352,18	2 647,82
18	7	2 647,82	1 000,00	8 352,18	1 647,82
19	8	1 647,82	1 000,00	9 352,18	647,82
20	9	647,82	647,82	10 000,00	0,00
21					
22					
23					

## 3 Implémentation dans Excel : Amortissement linéaire

### 3.1 Préparation

A	B	C	D	E	F
2	Données				
3	Valeur du bien		10 000,00		
4	Nb années		10		
5	Date d'acquisition		15/07/2016		11
6	Nb jours 1 <sup>re</sup> année		166		461,11
7	Taux		10,00%		1 000,00

Pour calculer :

- Nb jours 1<sup>re</sup> année
  - Les trois valeurs : "Valeur du bien, Nb années, Date d'acquisition" doivent être présentes,
  - il se calcule sur une durée de 360 jours pour un exercice de 12 mois.
  - $D6 = \text{SI}(\text{OU}(D3="" ; D4="" ; D5="" ) ; "" ; \text{JOURS360}(D5 ; \text{DATE}(\text{ANNEE}(D5) ; 12 ; 31)))$

— La valeur : "Nb jours 1<sup>re</sup> année" doit être présente (pour les 6 formules suivantes)

— Le Taux

- c'est l'inverse de la durée.
- $D7 = \text{SI}(D6="" ; "" ; \text{ARRONDI}(1/D4 ; 4))$

Trois cellules intermédiaires (E5 :E7) pour automatiser les formules. Pour calculer :

— Durée de l'amortissement :

- Si la 1<sup>re</sup> année est complète la durée = "Nb années" sinon la durée = "Nb années" + 1.
- $E5 = \text{SI}(D6="" ; "" ; \text{SI}(D6=360 ; D4 ; D4+1))$

— Amortissement :

- 1<sup>re</sup> année : "Valeur du bien" \* "Taux" \* "Nb jours 1<sup>re</sup> année" / 360
- $E6 = \text{SI}(D6="" ; "" ; \text{ARRONDI}(D3*D7*D6/360 ; 2))$  ou  $\text{ARRONDI}(E7*D6/360 ; 2)$
- Autres années : "Valeur du bien" \* "Taux"
- $E7 = \text{SI}(D6="" ; "" ; \text{ARRONDI}(D3*D7 ; 2))$
- Dernière année : le solde.

### 3.2 Création du tableau d'amortissement


	A	B	C	D	E	F
2		Données				
3		Valeur du bien		10 000,00		
4		Nb années		10		
5		Date d'acquisition		15/07/2016		11
6		Nb jours 1 <sup>re</sup> année		166		461,11
7		Taux		10,00%		1 000,00
8						
9						
10		Année	Base à amortir	Amortissement	Cumul	Valeur résiduelle
11		0				10 000,00
12		1	10 000,00	461,11	461,11	9 538,89

- Deux cellules intermédiaires (B11 et F11) pour automatiser les formules :
  - B11 =SI(D6="" ;"" ;0)
  - F11 =SI(D6="" ;"" ;D3)
- La valeur : "Année" doit être présente (pour les 5 formules suivantes)
- Colonne "Année" :
  - B12 =SI(B11="" ;"" ;SI(B11>=\$E\$5 ;"" ;B11+1))
- Colonne "Base à amortir" :
  - C12 =SI(B12="" ;"" ;\$F\$11)
- Colonne "Amortissement" :
  - 1<sup>re</sup> année : D12 =E6
  - Autres années : D12 =E7
  - Dernière année : Le solde ou la Valeur résiduelle précédente ce qui évite d'avoir une erreur d'arrondi (1 cent manquant lors de la dernière année)
  - D12 =SI(B12="" ;"" ;SI(B12=1 ;\$E\$6 ;SI(B12<\$E\$5 ;\$E\$7 ;F11)))
- Colonne "Cumul" :
  - E12 =SI(B12="" ;"" ;SOMME(\$D\$12 :D12))
- Colonne "Valeur résiduelle" :
  - F12 =SI(B12="" ;"" ;\$F\$11-E12)

La durée maximale d'un amortissement est fixée (pour notre exemple) à 30 ans. Il faut donc recopier les formules de B12 à F12 jusqu'à et y compris la ligne 42.

### 3.3 Mise en forme automatique des bordures

- Sélectionner B10 :F42
- choisir Accueil / Mise en forme conditionnelle
- choisir :
  - Nouvelle règle ...
  - Appliquer une mise en forme uniquement aux cellules contenant :
  - Aucune cellule vide
  - Format... - Bordure - Contour
- Valider



### 3.4 Remarques

- Masquer éventuellement les cellules E5 à E7
- Masquer éventuellement la ligne 11.
- Utiliser la Validation des données pour les cellules : "Valeur du bien, Nb années, Date d'acquisition".
- Utiliser aussi la Protection de ces cellules (et la Protection de la feuille).

### 3.5 Références

- Théorie : [www.coursdecomptabilite.be/theorie/1e001th005.php](http://www.coursdecomptabilite.be/theorie/1e001th005.php)
- Le classeur : [perso.unamur.be/~jmlamber/xl/amortissement/Amortissement.zip](http://perso.unamur.be/~jmlamber/xl/amortissement/Amortissement.zip)
- [Jean-Marie.Lambert@Skynet.be](mailto:Jean-Marie.Lambert@Skynet.be)
- [perso.unamur.be/~jmlamber/](http://perso.unamur.be/~jmlamber/)